

SIMPLIFICATIONS D'EXPRESSIONS

Maple dispose de tout un tas de fonctions pour factoriser, développer, linéariser, etc. Il n'est hélas pas toujours facile de savoir laquelle on doit utiliser chaque fois pour obtenir ce qu'on veut !

1 FONCTION simplify

Le fonction `simplify` simplifie. Mais qu'est-ce que simplifier ? Notion très variable qui dépend bien souvent du type de résultat que l'on attend : ce qu'on appelle « simplification » dans un calcul peut s'avérer « complication » dans un autre.

```
> cos(x)^2+sin(x)^2;
simplify(sin(x)^2+cos(x)^2);
```

$$\cos^2 x + \sin^2 x$$

$$1$$

```
> (a^2-b^2)/(a-b);
simplify((a^2-b^2)/(a-b));
```

$$\frac{a^2 - b^2}{a - b}$$

$$a + b$$

Retenez bien que les résultats obtenus ne sont pas toujours à la hauteur. Que penser par exemple de ceci ?

```
> simplify(sin(x)^3);
```

$$\sin x - \sin x \cos^2 x$$

On peut simplifier en choisissant le type de simplification qu'on souhaite à l'exclusion de tout autre. Les options principales pour nous sont `radical` pour les racines, `trig` pour les expressions trigonométriques et `exp` pour les exponentielles.

```
> a:=8^(1/3)*sin(x)^2*exp(3*ln(x)+2);
simplify(a);
simplify(a,radical);
simplify(a,exp);
simplify(a,radical,exp);
```

$$a := 8^{\frac{1}{3}} \sin^2 x e^{3 \ln x + 2}$$

$$2e^2 x^3 - 2e^2 x^3 \cos^2 x$$

$$2 \sin^2 x e^{3 \ln x + 2}$$

$$8^{\frac{1}{3}} e^2 x^3 \sin^2 x$$

$$2e^2 x^3 \sin^2 x$$

On peut demander à Maple d'effectuer des simplifications sous contraintes, i.e. des simplifications possibles parce qu'on fait des hypothèses supplémentaires. Premier exemple :

```
> simplify(x^2+y^2,{x+3*y=1});
```

$$10y^2 - 6y + 1$$

Dans cet exemple, Maple a choisi d'éliminer la variable y . Le même Maple sur le même ordinateur choisira une autre fois d'éliminer la variable x .

Deuxième exemple :

```
> sin(arcsin(t));
arcsin(sin(t));
simplify(arcsin(sin(t)));
simplify(arcsin(sin(t)),assume=RealRange(-Pi/2,Pi/2));
```

$$t$$

$$\text{Arcsin sin } t$$

$$\text{Arcsin sin } t$$

$$t$$

L'option « `assume=RealRange(-Pi/2,Pi/2)` » a débloqué Maple parce que la fonction `Arcsin` n'est pas la réciproque de sinus mais la réciproque de $\sin|_{[-\frac{\pi}{2}, \frac{\pi}{2}]}$.

2 FONCTION expand

La fonction `expand` est utile pour développer.

```
> expand((x+y)^5);
expand((t+1)*(t+3)/(t-1)/(t-2));      # Ne développe que le numérateur.
expand(sin(4*x)+3*cos(3*x));           # On dé-linéarise.
expand(exp(x+y));
expand(ln(x*y));
expand(n!*(n+1)!/(n-1)!^2);
```

$$\frac{x^5 + 5x^4y + 10x^3y^2 + 10x^2y^3 + 5xy^4 + y^5}{(t-1)(t-2)} + \frac{4t}{(t-1)(t-2)} + \frac{3}{(t-1)(t-2)}$$

$$8 \sin x \cos^3 x - 4 \sin x \cos x + 12 \cos^3 x - 9 \cos x$$

$$e^x e^y$$

$$\ln x + \ln y$$

$$n^2(n+1)$$

On peut demander à Maple de ne développer que certains termes comme l'illustrent les exemples qui suivent :

```
> expand((x+1)^3*(x+y)*(x-y)+(x+1)*(x-2), x+1);
expand((x-y)^2*cos(2*x+y), x-y);
```

$$\frac{(x+1)^3 x^2 - (x+1)^3 y^2 + (x+1)x - 2x - 2}{2(x-y)^2 \cos^2 x \cos y - (x-y)^2 \cos y - 2(x-y)^2 \sin x \cos x \sin y}$$

3 FONCTION factor

La fonction `factor` factorise — eh oui! — pour autant que c'est possible.

```
> factor(x^2+2*x+1);
factor(exp(x)*sin(x)+exp(x)*cos(y));
```

$$\frac{(x+1)^2}{e^x (\sin x + \cos y)}$$

4 FONCTION combine

La fonction `combine` regroupe, combine les termes des expressions qu'on lui passe en argument. En particulier elle linéarise les expressions trigonométriques. Parfois, sans qu'on sache trop pourquoi, il faut forcer Maple un peu au moyen d'une option : `trig`, `exp`, `ln`, `power`...

```
> combine(sin(x)*cos(x)^4-cos(x)*sin(3*x));      # On linéarise.
combine(3*ln(2)+4*ln(5/3));
combine(exp(x)*exp(y));
combine(x^m*x^n);                                # Bizarrement ceci ne produit rien...
combine(x^m*x^n,power);                          # ... mais l'option "power" est un succès.
```

$$\frac{1}{16} \sin(5x) + \frac{3}{16} \sin(3x) + \frac{1}{8} \sin x - \frac{1}{2} \sin(4x) - \frac{1}{2} \sin(2x)$$

$$\ln \frac{5000}{81}$$

$$e^{x+y}$$

$$x^m x^n$$

$$x^{m+n}$$

5 FONCTION collect

L'instruction « `collect(expr,var);` » regroupe les termes de l'expression `expr` en fonction des puissances de `var`. Le résultat est donc une expression polynomiale de variable `var`.

```
> collect(x^2*y^2+3*x^2*y+2*x*y^2-x^2+y^2-3*x+2,x);
expr:=expand(cos(3*x)+cos(x)*sin(3*x));
collect(expr,cos(x));
```

$$(y^2 + 3y - 1)x^2 + (2y^2 - 3)x + y^2 + 2$$

$$expr := 4 \cos^3 x - 3 \cos x + 4 \sin x \cos^3 x - \sin x \cos x$$

$$(4 + 4 \sin x) \cos^3 x + (-3 - \sin x) \cos x$$

6 FONCTION convert

La fonction `convert` convertit les expressions qu'on lui passe en argument en *autre chose*, cet autre chose dépendant directement des options qu'on impose. Elle permet beaucoup de conversions mais nous n'en présenterons que deux.

Pour transformer une expression quelconque en une expression à base de sinus et cosinus circulaires, on utilise l'option `sincos`; pour des tangentes, `tan`; pour des exponentielles, `exp`; pour des logarithmes, `ln`; pour des exponentielles et des logarithmes, `expln`; pour des fonctions sinus et cosinus circulaires et hyperboliques, `trig`; etc.

```
> convert(sin(x),tan);
convert(cos(x)*sinh(x),exp);
convert(arcsinh(x),ln);      # arcsinh est notre fonction Argsh.
```

$$\frac{2 \tan \frac{x}{2}}{1 + \tan^2 \frac{x}{2}}$$

$$\left(\frac{1}{2} e^{Ix} + \frac{1}{2} \frac{1}{e^{Ix}} \right) \left(\frac{1}{2} e^x - \frac{1}{2} \frac{1}{e^x} \right)$$

$$\ln \left(x + \sqrt{x^2 + 1} \right)$$

La fonction `convert` permet aussi de convertir un entier donné en base 10 en le même entier exprimé dans une autre base.

```
> convert(123456,base,10);      # Convertit 123456 en base 10.
convert(123456,base,7);
```

$$[6, 5, 4, 3, 2, 1]$$

$$[4, 3, 6, 2, 2, 0, 1]$$

Le dernier résultat signifie que $123456 = 1 \times 7^0 + 0 \times 7^1 + 2 \times 7^2 + 2 \times 7^3 + 6 \times 7^4 + 3 \times 7^5 + 4 \times 7^6$.